
Vim to TEXShop Preview Scripts

Nicola Vitacolonna

This note documents the procedure that enables two-way synchronisation between
Vim (MacVim) and TEXShop. The setup works for LATEX documents as well as ConTEXt
documents.

1 Backward (PDF to Source) Synchronisation

Three steps are needed:

1. Check Preferences > Source > Configure for External Editor.

2. To sync from TEXShop preview to MacVim, create a file othereditor in
/usr/local/bin with the following content:

#!/bin/sh
/usr/local/bin/mvim --remote-silent +"$1" "$2"

This will allow TEXShop to jump to line $1 of file $2 in MacVim when you
cmd-click somewhere in the preview. A similar command can be used
with regular Vim, provided that it has been compiled with +clientserver
(which is the default for MacVim).

3. Make the file executable:

chmod 755 /usr/local/bin/othereditor

2 Forward (Source to PDF) Synchronisation

To jump from a LATEX or ConTEXt source opened in (Mac)Vim to the correspond-
ing PDF preview in TeXShop, you need three Vim scripts. The first defines two
functions, one to open a PDF in TEXShop, and another that calls an AppleScript
script to jump to the point in the PDF that corresponds to a given line and column

in the source. The latter has two additional parameters to configure the synchro-
nisation method for LATEX or ConTEXt, respectively. This script should be saved in
~/.vim/autoload/tex.vim. This is the content of the script:

fun! tex#preview()
call job_start(['open', '-a', 'TeXShop.app', expand("%:p")])

endf

fun! tex#goto_texshop(path, linenr, colnr, context = 0, altpath = 0)
call job_start(['osascript',

\ '-e', 'tell front document of application "TeXShop"',
\ '-e', 'activate',
\ '-e', 'SyncWithOtherEditor',
\ '-e', a:altpath ? 'AlternateBinPath' : 'StandardBinPath',
\ '-e', a:context ? 'SyncConTeXt' : 'SyncRegular',
\ '-e', printf('sync_preview_line theLine %d', a:linenr),
\ '-e', printf('sync_preview_index theIndex %d', a:colnr),
\ '-e', printf('sync_preview_name theName "%s"', a:path),
\ '-e', 'end tell',
\ ])

endf

The other two scripts define buffer-local mappings (which you may change as you
like) to open a PDF in TEXShop and to jump to a given position in the PDF. For LATEX
save these mappings in ~/.vim/after/ftplugin/tex.vim:

nnoremap <silent><buffer> <leader>tv :<c-u>call tex#preview()<cr>
nnoremap <silent><buffer> <leader>ts

\ :<c-u>call tex#goto_texshop(expand("%:p"), line('.'), col('.'), 0, 0)<cr>

For ConTEXt, save these mappings in ~/.vim/after/ftplugin/context.vim:

nnoremap <silent><buffer> <leader>tv :<c-u>call tex#preview()<cr>
nnoremap <silent><buffer> <leader>ts

\ :<c-u>call tex#goto_texshop(expand("%:p"), line('.'), col('.'), 1, 1)<cr>

With the setup above, you may type \tv (where \ is Vim's leader) to open and view
in TEXShop the PDF document corresponding to the .tex file in the current buffer.
You may type \ts to jump to the position in the PDF corresponding to the current
cursor's position in Vim. The latter mapping assumes that the PDF file has already
been opened in TEXShop (with \tv), and it is the front document.

3 Further Configuration

It is recommended that you add some “magic lines” to your source documents. In
particular, you should specify the engine used to typeset your document:

% !TEX TS-program = <engine>

This is because, when TEXShop is configured to use an external editor, it will typeset
your document the first time it is opened. For LATEX documents, that should be all
you need. Read further if you use ConTEXt.

At the time of this writing (July 2021), the latest ConTEXt beta is needed for synchro-
nisation to work (the version in TEX Live 2021 will not work!). To work with the
latest beta proceed as follows:

Download and install ConTEXt as explained in ConTEXt Garden.

In TEXShop, set Preferences >Engine > Alternate Path to the path of your
ConTEXt beta installation.

In your source documents, add the following magic lines:

% !TEX useAlternatePath
% !TEX useConTeXtSyncParser

Make sure that SyncTEX is enabled in your source documents:

\setupsynctex[state=start,method=min]

4 External Resources

Using TEXShop with ConTEXt

Using ConTEXt in Vim

https://wiki.contextgarden.net/Installation
https://wiki.contextgarden.net/TeXShop
https://wiki.contextgarden.net/Vim

